IN-91-CR
123314
29P.

Semi-Annual Status Report

DEVELOPMENT OF AN ENGINEERING MODEL ATMOSPHERE FOR MARS

(SPECIAL TASK UNDER PROJECT ENTITLED "IMPROVEMENTS IN THE PERTURBATION SIMULATIONS OF THE GLOBAL REFERENCE ATMOSPHERIC MODEL")

C. G. Justus, Principal Investigator
George Chimonas, Co-Investigator
School of Geophysical Sciences
Atlanta, Georgia 30332-0340

March, 1988

Report Period: September 23, 1987 - March 22, 1988

Prepared for the

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
MARSHALL SPACE FLIGHT CENTER

Under Grant No. NAG8-078

Georgia Tech Project G-35-685

(NASA-CR-182462) DEVELOPMENT OF AN ENGINEERING MODEL ATMOSPHERE FOR MARS Semiannual Status Report, 23 Sep. 1987 - 22 Mar. 1988 (Georgia Inst. of Tech.) 29 p CSCL 03B G3/91

N88-17568

Unclas 0123314

Abstract

An engineering model atmosphere for Mars is being developed with many of the same features and capabilities of the highly successful Global Reference Atmospheric Model (GRAM) program for Earth's atmosphere (including mean values for density, temperature, pressure, and wind components, and density perturbation magnitudes and random perturbation profiles for density variations along specified trajectories). As an initial approach, the model is being built around the Martian atmosphere model computer subroutine (ATMOS) of Culp and Stewart (1984). In a longer-term program of research, additional refinements and modifications will be included. ATMOS includes parameterizations to simulate the effects of solar activity, seasonal variation, diurnal variation magnitude, dust storm effects, and effects due to the orbital position of Mars. One of the current shortcomings of ATMOS is the neglect of surface variation effects (with the exception of dust storm effects on the surface layer which are currently assumed to occur always with the same magnitude and at the same time every Martian "year"). The longer-term period of research and model building is to address some of these problem areas and provide further improvements in the model (including improved representation of near-surface variations, improved latitude-longitude gradient representation, effects of the large "annual" variation in surface pressure because of differential condensation/sublimation of the CO_2 atmosphere in the polar caps, and effects of Martian atmospheric wave perturbations on the magnitude of the expected density perturbations).

INTRODUCTION AND BACKGROUND

A highly successful and well-utilized engineering model for the Earth's atmosphere, the Global Reference Atmospheric Model (GRAM), has been developed at Georgia Tech (Justus, et al., 1980, 1986), and has undergone several improvement cycles. The GRAM program is used by several NASA Centers (and numerous other government agencies, industries and universities) for such projects as the Space Shuttle, the Space Station, Space Telescope, Tethered Satellite and AeroassistedOrbital Transfer Vehicle (AOTV). GRAM applications include orbital mechanics and lifetime studies, vehicle design and performance criteria, attitude control analysis problems, analysis of effects of short-term density variation from geomagnetic storms, and aerobraking analyses (for AOTV return from geosynchronous orbit to space-station rendezvous).

In addition to evaluating the mean density, temperature, pressure, and wind components at any height, latitude, longitude and monthly period, GRAM also allows for the simulation of "random perturbation" profiles about the mean conditions. This feature permits the simulation of a large number of realistic density profile realizations along the same trajectory through the atmosphere, with realistic values of scales of variation and peak perturbation values (e.g. the random perturbation profiles produce values which exceed the +3 standard deviation value approximately 1% of the time).

An engineering model atmosphere for Mars is being developed with many of the same features and capabilities of the GRAM program for Earth's atmosphere (including mean values for density, temperature, pressure, and wind components, and density perturbation magnitudes and random perturbation profiles for density variations along specified trajectories). As an initial approach, the model is being built around the Martian atmosphere model computer subroutine (ATMOS) of Culp and Stewart (1984). In a longer-term program of research and model-building, additional refinements and modifications will be included.

The ATMOS subroutine of Culp and Stewart incorporates results from a number of data and model sources, e.g. the Mars Reference Atmosphere (Kliore, 1982), occultation data and mass spectrometer data from Mariner and from Viking orbiters (Fjeldbo et al., 1966, 1970, 1977; Kliore et al., 1972; Stewart et al., 1972; Nier and McElroy, 1977), and data from the Viking lander atmospheric entry trajectories (Seiff and Kirk, 1977). ATMOS includes parameterizations to simulate the effects of solar activity, seasonal variation, diurnal variation magnitude, dust storm effects, and effects due to the orbital position of Mars.

One of the current shortcomings of ATMOS is the neglect of surface variation effects (with the exception of dust storm effects on the surface layer which are currently assumed to occur always with the same magnitude and at the same time every Martian "year").

PROGRAM OF RESEARCH

An engineering model atmosphere, similar to GRAM for the Earth's atmosphere, is being developed for Mars. Initially the model will be built around the ATMOS subroutine of Culp and Stewart (1984), which allows simulation of mean density, high and low density perturbation magnitudes, temperature and pressure at any altitude, areographic position and Martian season. Some features which have been added are methods to simulate mean wind components and random perturbation density profiles along specified trajectories.

Specific modifications and additions to the ATMOS subroutine and/or features of the new main program also include:

(1) Modification to latitude, longitude, and height input, and solar latitude, longitude, and Mars orbital radius input to the ATMOS subroutine. The original ATMOS routine required areocentric Cartesian coordinate (x-y-z) input values of these parameters. This change will make input to the program and operation of the program more similar to that for GRAM.

(2) A parameterization of the orbital parameters of Mars (from ephemeris data), so that these can be evaluated by the program, instead of input from separate ephemeris look-up results for each simulation case. The original ATMOS required the areocentric coordinates of the sun and the Julian date to be input. All of these are now evaluated from input of the calendar date (Earth-based Universal Time) and Greenwich time for which simulations are to be made. No solar or Martian orbital parameters are required among the input supplied by the user. The parameterization of the necessary orbital parameters was based on ephemeris data from 1976 through 1988, to provide appropriate accuracy during at least this period.

(3) Parameterizations for horizontal and vertical scales of density perturbations as a function of height have been developed, for use in the

random perbation model for density.

(4) Trajectory options (similar to GRAM) have been implemented whereby atmospheric parameters along linear steps in height, latitude and/or longitude can be simulated, or simulations can be made along any desired trajectory for which height-latitude-longitude positions are provided in an input data file.

(5) Simulation of mean wind components from mean pressure gradient and other factors has been added. Preliminary results from ATMOS indicate that above 100 km or so, the areostrophic winds (analogous to geostrophic winds, computed by a balance of pressure gradient and coriolis forces) become much too large. An areostrophic wind simulation technique including the damping effects of molecular viscosity (which becomes large at low density and high temperature) has therefore been included (similar to the viscous model at high altitudes recently added to the GRAM program).

All of these modifications and additions have been incorporated into an IBM-PC-DOS-compatible computer program (written in FORTRAN), the listing for which is given in the Appendix.

Figures 1 and 2 show sample output from the current MARS/ATMOS program. These are vertical profiles of density and temperature at the latitude-longitude position of the Viking 1 lander for the day of its landing. These may be compared with similar figures given by Culp and Stewart (1984).

PLANS FOR THE LONGER-TERM PROGRAM OF RESEARCH

One of the current shortcomings of ATMOS is the neglect of surface variation effects (with the exception of dust storm effects on the surface layer which are currently assumed to occur always with the same magnitude and at the same time every Martian "year"). The longer-term research and model development period will address some of these problem areas and provide further improvements in the model, including improved representation of near-surface variations, improved latitude-longitude gradient representation, and effects of Martian atmospheric wave perturbations. Specific areas of study and program improvements expected during this research and model-building period will include:

(1) Improvement in the treatment of surface and near-surface effects of seasonal and diurnal variations. Data from the Viking landers (Hess et al., 1977, 1980) will be analyzed and used, as well as results from theoretical studies of surface solar heating and expected resultant near-surface temperature variations and thermal wind patterns (e.g. Leovy and Mintz, 1969).

(2) Improvement in the latitude-longitude gradients of temperature, pressure, and density in the Martian troposphere (surface to about 40 km). Preliminary studies with ATMOS indicate that this routine significantly underestimates the latitude gradient of temperature and pressure (and hence the build up of tropospheric winds by the thermal wind gradient (vertical variation of the areostrophic wind).

(3) Inclusion of the effects of seasonal variations in the total mass of the Martian atmosphere, as different amounts of CO_2 are condensed at one pole and sublimed at the other, during the different Martian seasons. This effect is evident in the large ($\approx$ 30%) annual variations of surface pressure measured at the Viking Lander sites (Hess, et al., 1980). The present ATMOS routine uses a fixed value of surface pressure (6.1 mb), independent of season.

(4) Addition of time of year and magnitude selection for dust-storm perturbations. Presently ATMOS assumes a fixed magnitude dust storm which re-occurs during the same period of every Martian "year". The dust storm effects are currently treated by producing an isothermal layer at the Martian surface, with the higher thermal layers pushed upward by the same amount. The program will be modified so that the intensity of the dust-storm effect and the period of its occurrence can be selected by program input parameters.

A major activity of the extended period of research (to be conducted by co-investigator, Dr. George Chimonas) will be an analysis of the expected effects on density perturbation magnitudes due to wave phenomena in the Martian atmosphere. In the Earth's atmosphere, various wave phenomena (including planetary waves, tides, gravity waves, lee waves, etc.) are responsible for a substantial part of the atmospheric variability above the boundary layer. These perturbation magnitudes which can be added in with those already treated in the ATMOS routine. These magnitudes will have a direct effect on the random perturbation model for profiles of density perturbation values along the selected trajectories.

REFERENCES

Cornrath, B. J., 1981: "Planetary-Scale Wave Structure in the Martian Atmosphere", Icarus, 48: 246-255.

Culp, R.D. and A.I. Stewart, 1984: "Time-Dependent Model of the Martian Atmosphere fro Use in Orbit Lifetime and Sustenance Studies", J. Astronautical Sci., 32(3): 329-341.

Fjeldbo, G. et al., 1966: "Atmosphere of Mars: Mariner IV Models Compared", Science, 153: 1518-1523.

Fjeldbo, G. et al., 1970: "The Mariner 1969 Occultation Measurements of the Upper Atmosphere of Mars", Radio Science, 5: 381-386.

Fjeldbo, G. et al., 1977: "Viking Radio Occultation Measurements of the Martian Atmosphere and Topography: Primary Mission Coverage", J. Geophys. Res., 82: 4317-4324.

Gadian, A.M. and J.S.A. Green, 1983: "A Theoretical Study of Small Amplitude Waves in the Martian Lower Atmosphere and a Comparison Made with Those on Earth", Annales Geophysicae, 1: 239-244.

Hamilton, K. and R. R. Garcia, 1986: "Theory and Observations of the Short-Period Normal Mode Oscillations of the Atmosphere", J. Geophys. Res., 91(D11): 11,867-11,875.

Hess, S.L. et al., 1977: "Meteorological Results of the Surface of Mars: Viking 1 and 2", J. Geophys. Res., 82: 4459-4574.

Hess, S.L. et al., 1980: "The Annual Cycle of Pressure on Mars Measured by Viking Landers 1 and 2", Geophys. Res. Lett., 7: 197-200.

Justus, C. G., G. R. Fletcher, F. E. Gramling, and W. B. Pace, 1980: "The NASA/MSFC Global Reference Atmospheric Model - MOD 3 (With Spherical Harmonic Wind Model)", NASA CR 3256.

Justus, C. G., F. N. Alyea, and D. M. Cunnold, 1986: "Improvements in the Global Reference Atmospheric Model", Final Report on USRA Project P5042-OAO, Government Prime NAS8-36400/1).

Kliore, A. (editor), 1982: "The Mars Reference Atmosphere", Adv. Space. Res., 2.

Kliore, A. et al., 1972: "The Atmosphere of Mars from Mariner 9 Radio Occultation Measurements", Icarus, 17: 484-515.

PRECEDING PAGE BLANK NOT FILMED

Leovy, C.B., 1981: "Observations of Martian Tides Over Two Annual Cycles", J. Atmos. Sci., 38(1): 30.

Nier, A. O., and M. B. McElroy, 1977: "Composition and Structure of Mars' Upper Atmosphere: Results from Neutral Mass Spectrometer on Viking 1 and 2", J. Geophys. Res., 82: 4341-4350.

Pickersgill, A.O. and G.E. Hunt, 1979: "The Formation of Martian Lee Waves Generated by a Crater", J. Geophys. Res., 84: 8317-8331.

Pirraglia, J.A. and B.J. Cornrath, 1974: "Martian Tidal Pressure and Wind Fields Obtained from the Mariner 9 Infrared Spectroscopy Experiment", J. Atmos. Sci., 31(4): 318-329.

Seiff, A.L., and B.B. Kirk, 1977: "Structure of the Atmosphere of Mars in Summer at Mid-Latitudes", J. Geophys. Res., 82: 4364-4378.

Smith, S.A., D.C. Fritts, and T.E. Van Zandt, 1987: "Evidence of a Saturation Spectrum of Gravity Waves", J. Atmos. Sci., 44(12): 1404-1410.

Stewart, A., et al., 1972: "Mariner 9 Ultraviolet Spectrometer Experiment: Structure of Mars' Upper Atmosphere", Icarus, 17: 469-474.

Zurek, R.W., 1976: "Diurnal Tide in the Martian Atmosphere", J. Atmos. Sci., 33: 321.

Zurek, R.W., 1986: "Atmospheric Tidal Forcing of the Zonal-Mean Circulation: The Martian Dusty Atmosphere", J. Atmos. Sci., 43(7): 652.



Figure 1 - Simulated density profile (log-base-10 scale, units kg/m^3) from the MARS/ATMOS program, for a vertical path at the position of the Viking 1 Lander.



Figure 2 - As in Figure 1 for temperature.

APPENDIX A - MARS PROGRAM AND SUBROUTINE LISTINGS

```
C....   PROGRAM MARS - program to evaluate density, temperature,
C       pressure and wind components at a given time and position in
C       the Martian atmosphere
C.....................................................................
        IMPLICIT DOUBLE PRECISION (A-H,O-Z)
        DOUBLE PRECISION MARSAU,MARSRAD,NSWIND
        DIMENSION IDAY(12),IERR(10)
        character*8 lstfl,FILES(10),outfl
        EQUIVALENCE (IERR1,IERR(1)),(IERR2,IERR(2)),(IERR3,IERR(3)),
     &  (IERR4,IERR(4)),(IERR5,IERR(5)),(IERR6,IERR(6)),(IERR7,IERR(7))
     &  ,(IERR8,IERR(8)),(IERR9,IERR(9)),(IERR10,IERR(10))
        DATA files/'LIST','DENSLO','DENSAV','DENSHI','DENSRP','TEMP',
     &  'PRES','EWWIND','NSWIND','OUTPUT'/
        DATA IDAY/0,31,59,90,120,151,181,212,243,273,304,334/
        write(*,120)
  120   format(' Enter name for LIST file (CON for console listing): ')
        read(*,71)lstfl
 71     format(A)
        files(1)=lstfl
        WRITE(*,124)
 124    FORMAT(' Enter name for OUTPUT file:')
        read(*,71)outfl
        files(10)=outfl
        OPEN(8,file=lstfl,status='new',iostat=ierrl)
        OPEN(21,file='DENSLO',status='new',iostat=ierr2)
        OPEN(22,file='DENSAV',status='new',iostat=ierr3)
        OPEN(23,file='DENSHI',status='new',iostat=ierr4)
        OPEN(24,file='DENSRP',status='new',iostat=ierr5)
        OPEN(25,file='TEMP',status='new',iostat=ierr6)
        OPEN(26,file='PRES',status='new',iostat=ierr7)
        OPEN(27,file='EWWIND',status='new',iostat=ierr8)
        OPEN(28,file='NSWIND',status='new',iostat=ierr9)
        OPEN(29,file=outfl,status='new',iostat=ierrl0)
        do 87 j=1,10
 87     if(ierr(j).ne.0)goto 121
        goto 123
  121   write(*,122)files(j)
  122   format(a8,' Filename is old!')
        goto 9998
 123    DTR = 1.74532925199d-2
C...    VISCOSITY OF CARBON DIOXIDE: COEFFICIENTS FROM DATA IN 'TABLES
C       OF PHYSICAL AND CHEMICAL CONSTANTS' BY G.W.C. KAYE & T.H. LABY,
C       1961, LONGMANS, GREEN & CO., LONDON
        BETA = 1.578d-6
        SVAL = 2.40D2
C...    DAY = SIDEREAL DAY; CORFAC = CORIOLIS FACTOR (EXCEPT FOR
C       LATITUDE EFFECT)
        DAY = 2.4622962D1
        CORFAC = DTR/(1.0d1*DAY)
C...    FACTORS FOR USE IN CORRELATION FUNCTION
        AFAC = 19.51615854016301d0
        BFAC = 1.00041693941245578d0
```

```
200   WRITE(*,500)
500   FORMAT(' Enter Month, Day of Month, 4-digit Year,',
     & ' and Max Number Positions ')
      READ(*,*)MONTH,MDAY,MYEAR,NPOS
      IF(MONTH.LE.0.OR.MONTH.GT.12)GOTO 9999
      WRITE(*,300)
300   FORMAT(' Enter initial GMT Time (Hours, Minutes, Seconds)')
      READ(*,*)IHR,IMIN,SEC
      IF(MYEAR.LT.1900)MYEAR = MYEAR + 1900
      IF(NPOS.GT.0) GOTO 330
      OPEN(7,file='TRAJDATA',iostat=ierr7)
      if(ierr7.ne.0) goto 310
      goto 330
310   write(*,320)
320   format(' Unable to open Trajectory Data file')
      goto 9998
330   MAXNUM = NPOS - 1
      IF(NPOS.LE.0)MAXNUM = 32000
      NDAY = IDAY(MONTH) + MDAY
      IF(MOD(MYEAR,4).EQ.0.AND.MONTH.GT.2)NDAY = NDAY + 1
      XYEAR = (MYEAR - 1.9665d3)/4.0d0
      DATE = 2.439856d6 + 3.65d2*(MYEAR - 1.968d3) + NDAY +
     & DINT(XYEAR + DSIGN(0.5d0,XYEAR))
      WRITE(8,550)MONTH,MDAY,MYEAR,DATE,IHR,IMIN,SEC
550   FORMAT(' Date = ',I2,'/',I2,'/',I4,'  Julian Date = ',F9.0,
     & '  GMT Time = ',I2,':',I2,':',F4.1)
      DATE = DATE + IHR/2.4d1 + IMIN/1.440d3 + SEC/8.6400d4
      DATE0 = DATE
      WRITE(*,710)
710   FORMAT(/' Enter Starting Random Number (Any odd positive',
     & ' integer)')
      READ(*,*)NR1
      RHO = RAND(NR1)
      WRITE(8,720)NR1
720   FORMAT(' Starting random number = ',I6)
      WRITE(*,730)
730   FORMAT(/' Enter 1, 2 or 3 for output versus Height, Latitude',
     & ', or Longitude, respectively')
      READ(*,*)NVAR
      IF(NVAR.LT.1.OR.NVAR.GT.3)NVAR = 1
      IF(NPOS.LE.0)GO TO 750
10    WRITE(*,100)
100   FORMAT(' Enter Initial Height (km), Latitude (deg.),',
     & ' Longitude (deg.)')
      READ(*,*)FHGT,FLAT,FLON
      IF(FHGT.LT.0.)GOTO 200
20    WRITE(*,700)
700   FORMAT(' Enter Increments in Height (km), Latitude (deg.),',
     & ' Longitude (deg.),'/'  and Time (sec.)')
      READ(*,*)DELHGT,DELLAT,DELLON,DELTIME
750   WRITE(*,755)
755   FORMAT(' Computing data.')
      DO 900 I = 0,MAXNUM
      IF(NPOS.GT.0)GO TO 150
      READ (7,*,ERR=9998,END=200)CSEC,CHGT,CLAT,CLON
      DATE = DATE0 + CSEC/8.6400d4
```

```
      GO TO 160
150   CHGT = FHGT + I*DELHGT
      CLAT = FLAT + I*DELLAT
      CLON = FLON + I*DELLON
      CSEC = I*DELTIME
      DATE = DATE0 + CSEC/8.6400d4
160   IF(CHGT.LT.0.0)GO TO 200
      IF(DABS(CLAT).LE.90.d0)GO TO 170
      CLAT = DSIGN(1.80d2,CLAT) - CLAT
      CLON = CLON + 1.80d2
170   IF(CLON.LT.0.0)CLON = CLON + 3.60d2
      IF(CLON.GT.360.0) CLON = CLON - 3.60d2
      CALL ORBIT(DATE,SUNLAT,SUNLON,ALS,MARSAU)
      MARSRAD = MARSAU*1.496d+8
      CALL ATMOS(CHGT,CLAT,CLON,MARSRAD,SUNLAT,SUNLON,ALS,DATE,
     & HSCALE,TEMP,DENS,FACTHI,FACTLO,PRES,RSC)
      DENSLO = DENS*FACTLO
      DENSHI = DENS*FACTHI
      DPLUS = DENSHI - DENS
      DMINUS = DENS - DENSLO
      CALL NORMAL(Z1,Z2)
      CORREL = 0.
      VLS = 5.9d0 + 0.7d-01*CHGT**1.5d0
      IF(VLS.GT.CHGT.AND.CHGT.GT.1.35d2)VLS = CHGT
      IF(I.EQ.0)GO TO 180
      HLS = 1.350d3 + 9.0d0*CHGT
      IF(HLS.GT.6.750d3)HLS = 6.750d3
      DELNS = DTR*RSC*(CLAT - PLAT)/HLS
      DLON = CLON - PLON
      IF(DABS(DLON).GT.2.70d2)DLON = DSIGN(3.60d2,DLON) - DLON
      IF(DABS(DLON).GT.9.0d1)DLON = DLON - DSIGN(1.80d2,DLON)
      DELEW = DTR*RSC*DCOS(DTR*CLAT)*DLON/HLS
      DELZ = (CHGT - PHGT)/VLS
      DELX = DSQRT(DELNS**2 + DELEW**2 + DELZ**2)
      IF(DELX.LT.0.05d0)CORREL = 1.0d0 - AFAC*DELX**2
      IF(DELX.GE.0.05d0)CORREL = DEXP(-BFAC*DELX)
180   PHGT = CHGT
      PLAT = CLAT
      PLON = CLON
      RHO = CORREL*RHO + DSQRT(1.0d0 - CORREL**2)*Z1
      IF(RHO.LT.0.0)DENSP = DENS + RHO*DMINUS
      IF(RHO.GE.0.0)DENSP = DENS + RHO*DPLUS
      CALL ATMOS(CHGT,CLAT,CLON+2.5d0,MARSRAD,SUNLAT,SUNLON,ALS,DATE,
     & HLONP,TLONP,DLONP,FACTHI,FACTLO,PLONP,RLONP)
      CALL ATMOS(CHGT,CLAT,CLON-2.5d0,MARSRAD,SUNLAT,SUNLON,ALS,DATE,
     & HLONM,TLONM,DLONM,FACTHI,FACTLO,PLONM,RLONM)
      CLATP = CLAT + 2.5d0
      CLONP = CLON
      IF(CLATP.LE.9.0d1)GO TO 570
      CLATP = 1.80d2 - CLATP
      CLONP = CLONP + 1.80d2
570   CALL ATMOS(CHGT,CLATP,CLONP,MARSRAD,SUNLAT,SUNLON,ALS,DATE,
     & HLATP,TLATP,DLATP,FACTHI,FACTLO,PLATP,RLATP)
      CLATM = CLAT - 2.5d0
      CLONM = CLON
      IF(CLONM.GE.-9.0d1)GO TO 580
```

```
      CLATM = -1.80d2 - CLATM
      CLONM = CLONM + 1.80d2
580   CALL ATMOS(CHGT,CLATM,CLONM,MARSRAD,SUNLAT,SUNLON,ALS,DATE,
     & HLATM,TLATM,DLATM,FACTHI,FACTLO,PLATM,RLATM)
      DELNS = DTR*RSC*5.0d3
      WLAT = CLAT
      IF(DABS(WLAT).GT.8.75d1)WLAT = DSIGN(8.75d1,WLAT)
      IF(DABS(WLAT).LT.2.5d0)WLAT = DSIGN(2.5d0,WLAT)
      DELEW = DTR*RSC*DCOS(DTR*WLAT)*5.0d3
      CORIOL = CORFAC*DSIN(DTR*WLAT)
      VISC = BETA*TEMP**1.5d0/(TEMP + SVAL)
      VISCFAC = VISC/(1.0d6*DENS*VLS**2)
      DPDY = (PLATP-PLATM)/(DELNS*DENS)
      DPDX = (PLONP-PLONM)/(DELEW*DENS)
      DENOM = CORIOL**2 + VISCFAC**2
      EWWIND = (-CORIOL*DPDY - VISCFAC*DPDX)/DENOM
      NSWIND =  (CORIOL*DPDX - VISCFAC*DPDY)/DENOM
      WRITE(8,590)CSEC,CHGT,HSCALE,CLAT,CLON
590   FORMAT(' TIME (relative to initial time) = ',F12.1,' seconds'/
     & ' HEIGHT = ',F7.1,' km',13X,'SCALE HEIGHT = ',
     & F6.2,' km'/' LATITUDE = ',F8.3,' degrees     EAST LONGITUDE = ',
     & F8.3,' degrees')
      WRITE(8,600)TEMP,DENSLO,DENS,DENSHI,DENSP,PRES,
     & EWWIND,NSWIND
600   FORMAT(' TEMPERATURE = ',F7.1,' K'/
     & ' DENSITY (Low, Avg., High) = ',3G12.4,' kg/m**3'/
     & ' DENSITY PLUS PERTURBATION = ',G12.4,' kg/m**3'/
     & ' PRESSURE = ',G12.4,' N/m**2'/' EASTWARD WIND = ',F7.1,
     & ' m/s     NORTHWARD WIND = ',F7.1,' m/s')
      WRITE(8,650)
650   FORMAT(' ------------------------------------------------------',
     & '------------------')
      IF(NVAR.EQ.1)VAR = CHGT
      IF(NVAR.EQ.2)VAR = CLAT
      IF(NVAR.EQ.3)VAR = CLON
      WRITE(21,890)VAR,Dlog10(DENSLO)
      WRITE(22,890)VAR,Dlog10(DENS)
      WRITE(23,890)VAR,Dlog10(DENSHI)
      WRITE(24,890)VAR,Dlog10(DENSP)
      WRITE(25,890)VAR,TEMP
      WRITE(26,890)VAR,Dlog10(PRES)
      WRITE(27,890)VAR,EWWIND
      WRITE(28,890)VAR,NSWIND
890   FORMAT(F10.3,G12.4)
      SIGD = 5.0d1*(DENSHI-DENSLO)/DENS
      WRITE(29,895)CHGT,CLAT,CLON,Dlog10(DENS),TEMP,EWWIND,NSWIND,SIGD
895   FORMAT(F5.1,7F7.2)
900   CONTINUE
      GOTO 200
9998  STOP ' Error Termination Reading Data File'
9999  STOP ' Normal Termination'
      END
```

```
      subroutine orbit(xdate,ds,lon,ls,rad)
C...  Subroutine to compute Mars orbital parameters from input
C     Greenwich time (xdate, day and fraction).  Output consists of:
C     ds and lon - the areographic latitude and longitude of the sun,
C     ls - the heliocentric longitude of Mars (longitudinal angular
C     position of Mars around its orbital path, 0-360 degrees = one
C     Martian year), and
C     rad - the heliocentric orbital radius of Mars
C.....................................................................
      implicit double precision (a-h,o-z)
      double precision ls0,lon0,ls,lon
      data ls0,perls,TWOPI/6.36d0,6.8697964d2,6.28318530718d0/
      data lon0,perlon,fact/3.5758d2,1.02749118d0,0.9896538d0/
      data a0,a1,a2,a3,a4,a5,a6/1.5303331d0,.13661274d0,.38073649d-1,
     & -.34125165d-2,.56508078d-2,-.31633117d-3,-.33458978d-3/
      data b0,b1,b2,b3,b4,b5,b6/2.2057565d0,2.4703515d1,-1.5459399d0,
     & -0.50388237d0,2.2083870d0,-0.39583136d0,-0.077578018d0/
      data c0,c1,c2,c3,c4,c5,c6/-9.9300079d0,-2.8679217d0,1.0298739d1,
     & -0.53266474d0,-0.32173731d0,0.36206144d-1,-0.034765340d0/
      data d0,d1,d2,d3,d4,d5,d6/-3.5158255d0,6.5240771d0,-9.3284323d0,
     & 1.9052424d0,2.4887938d0,-0.49182825d0,0.10130513d-1/
      per1=6.87D2/TWOPI
      per2=6.96D2/TWOPI
      PI180 = TWOPI/3.60d2
      DATE = XDATE - 2.442779d6
      xls = ls0 + 3.60D2*date/perls
      xlon = lon0 + 3.60D2*(date - 2.922D3)/perlon
      TIME1=DATE/PER1
      TIME2=DATE/PER2
      LS=C0+C1*DSIN(TIME1)+C2*DCOS(TIME1)+C3*DSIN(2.*TIME1)
     & +C4*DCOS(2.*TIME1)+C5*DSIN(3.*TIME1)+C6*DCOS(3.*TIME1)+XLS
      DS=B0+B1*DSIN(TIME1)+B2*DCOS(TIME1)+B3*DSIN(2.*TIME1)
     & +B4*DCOS(2.*TIME1)+B5*DSIN(3.*TIME1)+B6*DCOS(3.*TIME1)
      LON=D0+D1*DSIN(TIME2)+D2*DCOS(TIME2)+D3*DSIN(2.*TIME2)
     & +d4*Dcos(2.*time2)+d5*Dsin(3.*time2)+d6*Dcos(3.*time2)
     & +xlon
      LS = DMOD(LS,3.6D2)
      LON = DMOD(LON,3.6D2)
      if(ls.lt.0.)ls = ls + 3.60D2
      if(lon.lt.0.)lon = lon + 3.60D2
      rad = a0 + a1*Dsin(time1) + a2*Dcos(time1) + a3*Dsin(2.*time1)
     & + a4*Dcos(2.*time1) + a5*DSin(3.*time1) + a6*Dcos(3.*time1)
      RETURN
      end


      SUBROUTINE ATMOS (CHGT,CLAT,CLON,MARSRAD,SUNLAT,SUNLON,ALS,
     & DATE,H,TEMP,DENST,UPFCTR,LWFCTR,PRES,RSC)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)

C
C     CHGT    HEIGHT OF SPACECRAFT ABOVE REFERENCE SURFACE (KM)(INPUT)
C     CLAT    LATITUDE OF SPACECRAFT (DEGREES) (INPUT)
C     CLON    LONGITUDE OF SPACECRAFT (DEGREES) (INPUT)
C     MARSRAD MARS ORBITAL RADIUS (KM) (INPUT)
```

```
C        SUNLAT  AEROCENTRIC LATITUDE OF SUN (DEGREES) (INPUT)
C        SUNLON  AEROCENTRIC LONGITUDE OF SUN (DEGREES) (INPUT)
C        ALS     MARS LONGITUDE IN ORBIT (INPUT)
C        DATE    JULIAN DATE (INPUT)
C        H       SCALE HEIGHT AT SPACECRAFT POSITION (KM) (OUTPUT)
C        TEMP    TEMPERATURE AT SPACECRAFT POSITION (K) (OUTPUT)
C        DENST   MASS DENSITY AT SPACECRAFT POSITION (KG/M**3) (OUTPUT)
C        UPFCTR  UPPER DEVIATION FACTOR ON MASS DENSITY  (OUTPUT)
C        LWFCTR  LOWER DEVIATION FACTOR ON MASS DENSITY  (OUTPUT)
C        PRES    PRESSURE AT SPACECRAFT POSITION (N/M**2) (OUTPUT)
C        RSC     AEROCENTRIC RADIUS TO SPACECRAFT (KM) (OUTPUT)
C********************************************************************
C        A       SEMIMAJOR AXIS OF MARS ORBIT (KM)
C        AH      AVERAGE SCALE HEIGHT WHICH IS FUNCTION OF TINF (KM)
C        AF      MAJOR AMPLITUDE OF SOLAR WAVE
C        AVOGDR  RECIPROCAL OF AVOGADROS NUMBER (KG-MOLE)
C        B       MINOR AMPLITUDE OF SOLAR WAVE
C        DL      LONGITUDE DIFFERENCE BETWEEN SPACECRAFT AND SUN
C        DT      DIFFERENCE OF DATE BETWEEN PRESENT AND REFERENCE
C        DTR     CONSTANT CONVERTED DEGREE TO RADIAN
C        DZDS    INCREMENT OF ALTITUDE DUE TO DUST STORM (KM)
C        DZSN    INCREMENT OF ALTITUDE DUE TO SUN (KM)
C        EB      EXPONENTIAL CONSTANT FOR ZB < Z < ZF
C        EL      EXPONENTIAL CONSTANT FOR ZL < Z < ZA
C        ED      AMPLITUDE OF DIURNAL EFFECT  (KM)
C        ES      AMPLITUDE OF SEASONAL EFFECT (KM)
C        F(I)    VECTOR STORED FRACTION OF EACH GAS AT ZF
C        F107    SOLAR FLUX AT 10.7 CENTIMETER WAVELENGTH
C        FP      FRACTION OF HELIOCENTRIC PERIOD
C        G6      GRAVITY AT ALTITUDE Z6 (CM/SEC**2)
C        GL      GRAVITY AT ALTITUDE ZL (CM/SEC**2)
C        GA      GRAVITY AT ALTITUDE ZA (CM/SEC**2)
C        GB      GRAVITY AT ALTITUDE ZB (CM/SEC**2)
C        GF      GRAVITY AT ALTITUDE ZF (CM/SEC**2)
C        GFOKT   PARAMETER EQUALS 10.*GF/(K*TINF)
C        H6      RECIPROCAL OF SCALE HEIGHT FOR Z6 < Z < ZL (1/KM)
C        HA      RECIPROCAL OF SCALE HEIGHT FOR ZA < Z < ZB (1/KM)
C        HINF    RECIPROCAL OF SCALE HEIGHT FOR HIGH ALTITUDE (1/KM)
C        IC      PARAMETER USED FOR SWITCHING TINF
C        K       UNIVERSAL GAS CONSTANT (J/KG-MOLE * K)
C        MBAR    AVERAGE MOLECULAR WEIGHT OF GASES
C        MBOK    PARAMETER EQUALS 10.*(MBAR/K)
C        MW(I)   VECTOR STORED MOLECULAR WEIGHT OF EACH GAS
C        NC      NUMBER OF CONSTITUENT GASES
C        NBDNS(I)     VECTOR STORED NUMBER DENSITY OF EACH GAS (1/CM**3)
C                WHERE      I - 1 REPRESENTS CO2
C                                 2 REPRESENTS O
C                                 3 REPRESENTS N2
C                                 4 REPRESENTS AR
C                                 5 REPRESENTS CO
C                                 6 REPRESENTS O2
C                                 7 REPRESENTS HE
C                                 8 REPRESENTS H
C                                 9 REPRESENTS H2
C        P       PRESSURE AT SPACECRAFT POSITION (N/M**2)
C        P6      PRESSURE AT Z6 EQUALS 6.1 MILLIBAR
```

```
C        PA       PRESSURE AT ZA (N/M**2)
C        PB       PRESSURE AT ZB (N/M**2)
C        PL       PRESSURE AT ZL (N/M**2)
C        PF       PRESSURE AT ZF (N/M**2)
C        PP(I)    VECTOR STORED PARTIAL PRESSURE OF EACH GAS
C        PERIOD   SIDEREAL PERIOD OF MARS (DAYS)
C        R61MB    AREOCENTRIC DISTANCE OF 6.1-MILLIBAR SPHEROID (KM)
C        RA       SEMIMAJOR AXIS OF ATMOSPHERIC ELLIPSOID (KM)
C        RB       SEMIMINOR AXIS OF ATMOSPHERIC ELLIPSOID (KM)
C        RC       POLAR AXIS OF ATMOSPHERIC ELLIPSOID (KM)
C        REFDAT   DATE OF MARS PERIHELION
C        RSC      AREOCENTRIC DISTANCE OF SPACECRAFT (KM)
C        RSN      HELIOCENTRIC DISTANCE OF MARS (KM)
C        RSQ      DUMMY VARIABLE FOR CALCULATING R61MB
C        SCLAT    LATITUDE OF SPACECRAFT (RAD.) (ROTATING COORDINATES)
C        SCLONG   LONGITUDE OF SPACECRAFT (RAD.)
C        SNLAT    LATITUDE OF SUN  (RAD.) (ROTATING COORDINATES)
C        SNLONG   LONGITUDE OF SUN (RAD.)
C        SIGMA    ESTIMATED STANDARD DEVIATION PARAMETER
C        SOLDAY   DAYS IN EARTH YEAR
C        TF       TEMPERATURE AT THERMOSPHERIC SURFACE (K)
C        T        TEMPERATURE AT SPACECRAFT POSITION (K)
C        T6       TEMPERATURE AT ALTITUDE Z6
C        TA       TEMPERATURE AT ALTITUDE ZA    (K)
C        TB       TEMPERATURE AT ALTITUDE ZB
C        TIME     LOCAL TIME AT SPACECRAFT POSITION (RAD.)
C        TPL      RATE OF CHANGE OF TEMP. FOR ZL < Z < ZA (K/KM)
C        TPB      RATE OF CHANGE OF TEMP. FOR ZB < Z < ZF (K/KM)
C        TINF     EXOSPHERIC TEMPERATURE (K)
C        U        UNIVERSAL GRAVITATIONAL CONSTANT (CM*KM**2/SEC**2)
C        VD( )    VECTOR STORED DENSITY
C        X        AREOPOTENTIAL ALTITUDE (KM)
C        XL       AREOPOTENTIAL ALTITUDE AT Z=ZL
C        XA       AREOPOTENTIAL ALTITUDE AT Z=ZA   (KM)
C        XB       AREOPOTENTIAL ALTITUDE AT Z=ZB
C        XF       AREOPOTENTIAL ALTITUDE AT Z=ZF
C        Z        ALTITUDE OF SPACECRAFT ABOVE REFERENCE SURFACE (KM)
C        Z6       ALTITUDE OF 6.1 MB PRESSURE, EQUAL ZERO (KM)
C        ZL       BEGINNING ALTITUDE OF TROPOSPHERE (KM)
C        ZA       BEGINNING ALTITUDE OF STRATOSPHERE  (KM)
C        ZB       ALTITUDE OF END OF STRATOSPHERE  (KM)
C        ZF       ALTITUDE OF THERMOSPHERIC SURFACE (KM)
C ******************************************************************
      DIMENSION PP(9),TINF(3),ZF(3),F(9),DZDS(3),VD(3)
      DOUBLE PRECISION MB,MBAR,MBOK,NBDNS(9),MW(9),K,LWFCTR,MARSRAD
      DATA A,AF,AVOGDR,ED,ES/2.279D8,6.0D1,0.1660081676D-20,.1d0,.1d0/
      DATA F/.93d0,.2d-1,.27d-1,.16d-1,.1d-1,.13d-2,5.D-5,1.0D-6,4.0D
     & -6/
      DATA K,MW/8.31439d3,44.01d0,1.6d1,28.012d0,39.948d0,28.01d0,
     & 3.2d1,4.0d0,1.0d0,2.0d0/
      DATA MBAR,NC,PERIOD,PI,PI2/43.3d0,9,6.8698d2,3.14159265d0,
     & 6.28318531d0/
      DATA P6,RA,RB,RC,DTR/6.10D2,3.39467D3,3.39321D3,3.37678D3,
     & 0.17453292D-1/
      DATA REFDAT,SIGMA,SOLDAY,U/2.443951D6,1.d0,3.6525636D2,
     & .42828443D10/
```

```
C ... CONVERT COORDINATES SYSTEM .....
c       write(*,33)chgt,clat,clon,marsrad
c       write(*,44)sunlat,sunlon,als,h
c       write(*,45)upfctr,lwfctr,pres,rsc
 33     format(' chgt=',g12.4,' clat=',g12.4,' clon=',g12.4,
     & ' marsrad=',g12.4)
 44     format(' sunlat=',g12.4,' sunlon=',g12.4,' als=',g12.4,
     & ' h=',g12.4)
 45     format(' upfctr=',g12.4,' lwfctr=',g12.4,' pres=',g12.4,
     & ' rsc=',g12.4,/)
        SNLONG = DTR*SUNLON
        SNLAT = DTR*SUNLAT
        SCLAT = DTR*CLAT
        SCLONG = DTR*CLON
        RSN = MARSRAD
        ELON = DTR*(CLON + 1.085D2)
        COSLAT = DCOS(SCLAT)
        IF(COSLAT.LE.0.0)COSLAT = 0.0
        RSQ = (COSLAT*DCOS(ELON)/RA)**2 + (COSLAT*DSIN(ELON)/RB)**2
     & + (DSIN(SCLAT)/RC)**2
        R61MB = DSQRT(1.0D0/RSQ)
        RSC = R61MB + CHGT
C ... CALCULATE LOCAL TIME .....
        DL=SCLONG-SNLONG
        TIME=PI+DL
        IF( TIME.LT.0. ) TIME=TIME+PI2
        IF( TIME.GE.PI2 ) TIME=TIME-PI2
        SCLONG=SCLONG+1.085D2*DTR
C ... CALCULATE THE FRACTION OF HELIOCENTRIC PERIOD .....
        DT=DATE-REFDAT
        FP=DMOD(DT,PERIOD)/PERIOD
        IF( FP.LT.0. ) FP=1.d0+FP
        DAYS=DMOD(DATE-2.443935D6,2.635D1)
        IF( DAYS.LT.0. ) DAYS=DAYS+2.635d1
        AOR = A/RSN
        R61MB=R61MB-2.8D0*DSIN((ALS-5.9D1)*DTR)
C ... CALCULATE THE INCREMENT OF DUST STORM .....
        DL1=ALS-2.10D2
        DL2=ALS-2.70D2
        IF( DL1.LT.0. ) DL1=DL1+3.60D2
        IF( DL2.LT.0. ) DL2=DL2+3.60D2
        DZ1=7.5D0*(1.0D0-DEXP(-DL1/5.0D0))*DEXP(-DL1/35.0D0)/6.5D-01
        DZ2=1.35D1*(1.0D0-DEXP(-DL2/5.0D0))*DEXP(-DL2/35.0D0)/6.5D-01
        DZDS(1)=DZ1+DZ2+1.2d0
C ... CALCULATE THE DIURNAL AND SEASONAL EFFECT .....
        DZSN=ED*DCOS(TIME-3.6651914d0)*DCOSLAT+ES*DCOS(SNLAT-SCLAT)
C ... CALCULATE F10.7 AND TEMPERATURE .....
        Y=(DATE-2.442870D6)*PI2/(1.104D1*SOLDAY)
        F107=7.5D1+AF*(1.0D0-DCOS(Y+3.0D1*DTR*(1.0D0-DCOS(Y))))
        B=1.1D-03*F107
        F107=F107*(1.0D0+B*DSIN(PI2*DAYS/2.635D1))
        T6=2.20D2*AOR
        TA=1.40D2*AOR
        TB=TA
        TF=1.725D2*AOR
C ... CALCULATE PARAMETERS AT POINTS L,A,B,F .....
```

```
      F(2)=2.0D-02*(1.0D0-0.5D0*DSIN(TIME)*DSQRT(COSLAT))
      F(1)=0.9457D0-F(2)
      Z6=0.
      DZDS(2)=DZDS(1)+3.0d0*SIGMA
      DZDS(3)=DZDS(1)-3.0d0*SIGMA
      IF( DZDS(3).LT.0. ) DZDS(3)=0.
      TINF(1)=2.05D2*AOR*AOR*((F107+7.5D1)/1.50D2)
      TINF(2)=TINF(1)*(1.d0+0.16d0*SIGMA)
      TINF(3)=TINF(1)*(1.d0-0.16d0*SIGMA)
      ZF(1)=DZDS(1)+1.25D2*AOR + DZSN
      ZF(2)=ZF(1)-3.0d0*SIGMA
      ZF(3)=ZF(1)+3.0d0*SIGMA
      MBOK=1.0D1*MBAR/K
      G6=U/R61MB**2
      H6=MBOK*G6/T6
      IC=0
      Z=RSC-R61MB
      G=U/(RSC*RSC)
      DO 10 M=1,3
        ZL=DZDS(M)/(1.d0-TA/T6)
        ZA=ZL+3.2D1*AOR
        ZB=DZDS(M)+1.0D2*AOR
        GL=U/(R61MB+ZL)**2
        GA=U/(R61MB+ZA)**2
        GB=U/(R61MB+ZB)**2
        XL=(ZL-Z6)*(R61MB+Z6)/(R61MB+ZL)
        XA=(ZA-ZL)*(R61MB+ZL)/(R61MB+ZA)
        XB=(ZB-ZA)*(R61MB+ZA)/(R61MB+ZB)
        TPL=(TA-T6)/XA
        EL=MBOK*GL/TPL
        HA=MBOK*GA/TA
        PL=P6*DEXP(-XL*H6)
        PA=PL*(T6/TA)**EL
        PB=PA*DEXP(-XB*HA )
        GF=U/(R61MB+ZF(M))**2
        XF=(ZF(M)-ZB)*(R61MB+ZB)/(R61MB+ZF(M))
        TPB=(TF-TB)/XF
        EB=MBOK*GB/TPB
        PF=PB*(TB/TF)**EB
        AH=TINF(M)/18.0D0
        GFOKT=1.0D1*GF/(K*TINF(M))
        F(7)=TINF(M)*6.3D-9
        TSQ= DSQRT(TINF(M))
        F(9)=TSQ*1.44D-7
        F(8)=TINF(M)*DEXP(-0.5D0*TSQ)*4.0D-5
        IF( TINF(M).GE.400.) F(8)=TSQ*4.6D-9*DEXP(1.44D3/TINF(M))/
     *                            (1.0D0+1.44D3/TINF(M))
C
C ... CALCULATE MASS DENSITY FOR Z < ZF .....
        IF( Z.GT.ZF(M) ) GO TO 5
C             FOR ALTITUDE BETWEEN ZB 6ND ZF  .....
        IF( Z.LT.ZB ) GO TO 1
        X=(Z-ZB)*(R61MB+ZB)/(R61MB+Z)
        T=TB+X*TPB
        P=PB*(TB/T)**EB
        GO TO 4
```

```
C             FOR ALTITUDE BETWEEN ZA AND ZB  .....
    1   IF( Z.LT.ZA ) GO TO 2
          T=TA
          X=(Z-ZA)*(R61MB+ZA)/(R61MB+Z)
          P=PA*DEXP(-X*HA )
          GO TO 4
C             FOR ALTITUDE BETWEEN ZL AND ZA  .....
    2   IF( Z.LT.ZL ) GO TO 3
          X=(Z-ZL)*(R61MB+ZL)/(R61MB+Z)
          T = T6 + X*TPL
          P = PL*(T6/T)**EL
          GO TO 4
C             FOR ALTITUDE BETWEEN Z6 AND ZL  .....
    3   X=(Z-Z6)*R61MB/(R61MB+Z)
          T=T6
          P = P6*DEXP(-X*H6)
    4   DENST = 0.1d0*MBOK*P/T
          MB=MBAR
          GO TO 9
C
C ...   CALCULATE MASS DENSITY FOR ALTITUDE Z > ZF  .....
    5   X=(Z-ZF(M))/(1.d0+ (Z-ZF(M))/(RSC-Z+ZF(M)) )
          T=TINF(M)-(TINF(M)-TF)*DEXP(-X/AH)
          TFOT=TF/T
          TKAV=1.0d0/(T*K*AVOGDR)
          DO 6 I=1,NC
            HINF=MW(I)*GFOKT
            PP(I)=PF*F(I)*DEXP(-X*HINF)*TFOT**(AH*HINF)
            NBDNS(I)=PP(I)*TKAV
    6     CONTINUE
          SNM2=0.
          P=0.
          DENST=0.
          DO 7 I=1,NC
            P=P+PP(I)
            SNM2=SNM2+NBDNS(I)*MW(I)*MW(I)
            DENST=DENST+NBDNS(I)*MW(I)
            IF( I.EQ.7 ) DNST7=DENST
    7     CONTINUE
C
          IF((DENST-DNST7).LT.(50.*DNST7) .OR. IC.EQ.1 ) GO TO 8
          IF( Z.LT.500. ) GO TO 8
          IC=1
          TINFO=TINF(2)
          TINF(2)=TINF(3)
          TINF(3)=TINFO
    8   CONTINUE
          MB=SNM2/DENST
          DENST=DENST*AVOGDR
C
    9   VD(M)=DENST
          IF( M.NE.1 ) GO TO 10
          TEMP=T
        PRES = P
          H=0.1d0*K*T/(MB*G)
   10   CONTINUE
```

```
C
      DENST=VD(1)
      UPFCTR=VD(2)/VD(1)
      LWFCTR=VD(3)/VD(1)
      IF( Z.GT.ZA ) RETURN
      UPFCTR=1.d0+(0.1d0+ 1.45d0*Z/ZB ) * SIGMA
      LWFCTR=0.9d0 - SIGMA*0.145D-1*Z/(ZA-ZL)
      return
      END



      function rand(i)
C...  Function to compute random number uniformly distributed
C     between 0 and 1
C.................................................................
      double precision x,rand
C...  static x
      if(i.ne.0) x = i/2.62144D+05
      x = x*5.09D2
      x = x - Dint(x)
      rand = x
      return
      end


      subroutine normal(d1,d2)
C...  SUBROUTINE to compute a pair of normally (Gaussian) distributed
C     variables with mean value 0 and standard deviation 1
C.................................................................
      implicit double precision (a-h,o-z)
 50   x = rand(0)
      y = 2.D0*rand(0) - 1.
      xx = x**2
      yy = y**2
      s = xx + yy
      if(s-1)51,51,50
 51   al = dsqrt(-2.*dlog(rand(0)))/s
      d1 = (xx - yy)*al
      d2 = 2.D0
      return
      end
```

APPENDIX B - DOCUMENTATION FOR THE MARS/ATMOS PROGRAM

The MARS/ATMOS program is invoked in interactive mode by entering MARS. An example of the program interactive operation is shown as Table B-1. The user may select filenames (or CON for console output) for a LIST file and an OUTPUT file. Examples of the LIST and OUTPUT files are shown in Tables B-2 and B-3. The LIST file has self-explanatory labels. The OUTPUT file contains data suitable for reading into auxiliary programs, and consists of: height (km), latitude (degrees), longitude (degrees), log-base-10 of density (kg/m^3), temperature (K), eastward wind (m/s), northward wind (m/s), and density perturbation magnitude (%, defined as high density minus low density divided by twice the average density).

Other output files, suitable for plotting, contain x, y pairs of values, where x may be selected as height, latitude or longitude and y is either log-base-10 of low density (DENSLO file), log-base-10 of average density (DENSAV file), log-base-10 of high density (DENSHI file), log-base-10 of density-plus-perturbation (DENSRP file), temperature (TEMP file), pressure (PRES file), eastward wind (EWWIND file), or northward wind (NSWIND file). A sample of the DENSAV file is shown in Table B-4.

The user is prompted (see Table B-1) to enter the month, day of month, year and maximum number of positions to compute. If the maximum number of positions is input as zero, position data (time, height, latitude, longitude) is read from a file called TRAJDATA. Otherwise the program steps linearly through height, latitude, and/or longitude for the desired number of positions. Greenwich Mean time for the starting position in input in hours, minutes and seconds. All solar and Mars orbital positions are computed by the program from the date and time input.

The program computes random perturbations in density, relative to the average density and with perturbation magnitudes evaluated from the low and high density values. A starting random number (any odd positive integer) must be input to start the random perturbation sequence. If a different number is selected, a different sequence will be generated; if the same starting random number is selected, the same random sequence will be selected.

A parameter value of 1, 2 or 3 is input to select plot output files (DENSAV file, etc.) versus height, latitude, or longitude respectively. For plotting variables versus height, option 1 is selected (see Table B-4).

Initial height (km), latitude (degrees, N=+, S=-) and longitude (0-360° E) must be input, along with increments of height, latitude, longitude and time for the program-generated position sequence. Increments may be positive (increasing value with position step) or negative (decreasing value with position step).

The program is terminated by entering zero or negative values for initial date and number of positions.

TABLE B-1

SAMPLE INTERACTIVE OPERATION OF MARS/ATMOS PROGRAM

```
 MARS
Enter name for LIST file (CON for console listing):
MARSLIST
Enter name for OUTPUT file:
MARSOUT
Enter Month, Day of Month, 4-digit Year, and Max Number Positions
7 20 1976 13
Enter initial GMT Time (Hours, Minutes, Seconds)
0 0 0

Enter Starting Random Number (Any odd positive integer)
123

Enter 1, 2 or 3 for output versus Height, Latitude, or Longitude, respectively
1
Enter Initial Height (km), Latitude (deg.), Longitude (deg.)
600 22.23 312.07
Enter Increments in Height (km), Latitude (deg.), Longitude (deg.),
 and Time (sec.)
-50 0 0 0
Computing data.
Enter Month, Day of Month, 4-digit Year, and Max Number Positions
0 0 0 0
 Normal Termination
```

TABLE B-2

PRINTOUT OF MARSLIST LIST FILE

```
Date = 7/20/1976  Julian Date =  2442980.  GMT Time =  0: 0:  .0
Starting random number =    123
TIME (relative to initial time) =          .0 seconds
HEIGHT =   600.0 km              SCALE HEIGHT = 428.53 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
TEMPERATURE =   168.7 K
DENSITY (Low, Avg., High) =    .1513E-15   .3224E-15   .9169E-15 kg/m**3
DENSITY PLUS PERTURBATION =    .1597E-14 kg/m**3
PRESSURE =    .4204E-09 N/m**2
EASTWARD WIND =       .0 m/s     NORTHWARD WIND =       .5 m/s
------------------------------------------------------------------------
TIME (relative to initial time) =          .0 seconds
HEIGHT =   550.0 km              SCALE HEIGHT = 354.96 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
TEMPERATURE =   168.7 K
DENSITY (Low, Avg., High) =    .2795E-15   .3659E-15   .6832E-15 kg/m**3
DENSITY PLUS PERTURBATION =    .9299E-15 kg/m**3
PRESSURE =    .4671E-09 N/m**2
EASTWARD WIND =       .0 m/s     NORTHWARD WIND =       .5 m/s
------------------------------------------------------------------------
TIME (relative to initial time) =          .0 seconds
HEIGHT =   500.0 km              SCALE HEIGHT = 220.23 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
TEMPERATURE =   168.7 K
DENSITY (Low, Avg., High) =    .3199E-15   .4357E-15   .1002E-14 kg/m**3
DENSITY PLUS PERTURBATION =    .1632E-14 kg/m**3
PRESSURE =    .5225E-09 N/m**2
EASTWARD WIND =       .0 m/s     NORTHWARD WIND =       .4 m/s
------------------------------------------------------------------------
TIME (relative to initial time) =          .0 seconds
HEIGHT =   450.0 km              SCALE HEIGHT =  96.39 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
TEMPERATURE =   168.7 K
DENSITY (Low, Avg., High) =    .3775E-15   .6165E-15   .2109E-14 kg/m**3
DENSITY PLUS PERTURBATION =    .3254E-14 kg/m**3
PRESSURE =    .5954E-09 N/m**2
EASTWARD WIND =       .0 m/s     NORTHWARD WIND =       .4 m/s
------------------------------------------------------------------------
TIME (relative to initial time) =          .0 seconds
HEIGHT =   400.0 km              SCALE HEIGHT =  45.41 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
TEMPERATURE =   168.7 K
DENSITY (Low, Avg., High) =    .5169E-15   .1367E-14   .6605E-14 kg/m**3
DENSITY PLUS PERTURBATION =    .8583E-14 kg/m**3
PRESSURE =    .7277E-09 N/m**2
EASTWARD WIND =      -.1 m/s     NORTHWARD WIND =       .4 m/s
------------------------------------------------------------------------
TIME (relative to initial time) =          .0 seconds
HEIGHT =   350.0 km              SCALE HEIGHT =  31.88 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
```

TEMPERATURE = 168.7 K
DENSITY (Low, Avg., High) = .1239E-14 .5334E-14 .2647E-13 kg/m**3
DENSITY PLUS PERTURBATION = .4095E-13 kg/m**3
PRESSURE = .1153E-08 N/m**2
EASTWARD WIND = -.6 m/s NORTHWARD WIND = .6 m/s

TIME (relative to initial time) = .0 seconds
HEIGHT = 300.0 km SCALE HEIGHT = 28.20 km
LATITUDE = 22.230 degrees EAST LONGITUDE = 312.070 degrees
TEMPERATURE = 168.7 K
DENSITY (Low, Avg., High) = .6768E-14 .2875E-13 .1205E-12 kg/m**3
DENSITY PLUS PERTURBATION = .1240E-12 kg/m**3
PRESSURE = .3283E-08 N/m**2
EASTWARD WIND = -2.7 m/s NORTHWARD WIND = 1.6 m/s

TIME (relative to initial time) = .0 seconds
HEIGHT = 250.0 km SCALE HEIGHT = 25.69 km
LATITUDE = 22.230 degrees EAST LONGITUDE = 312.070 degrees
TEMPERATURE = 168.7 K
DENSITY (Low, Avg., High) = .5611E-13 .1827E-12 .6238E-12 kg/m**3
DENSITY PLUS PERTURBATION = .6101E-12 kg/m**3
PRESSURE = .1642E-07 N/m**2
EASTWARD WIND = -11.4 m/s NORTHWARD WIND = 6.1 m/s

TIME (relative to initial time) = .0 seconds
HEIGHT = 200.0 km SCALE HEIGHT = 17.80 km
LATITUDE = 22.230 degrees EAST LONGITUDE = 312.070 degrees
TEMPERATURE = 168.7 K
DENSITY (Low, Avg., High) = .6587E-12 .1725E-11 .5145E-11 kg/m**3
DENSITY PLUS PERTURBATION = .4576E-11 kg/m**3
PRESSURE = .1221E-06 N/m**2
EASTWARD WIND = -41.1 m/s NORTHWARD WIND = 34.7 m/s

TIME (relative to initial time) = .0 seconds
HEIGHT = 150.0 km SCALE HEIGHT = 10.22 km
LATITUDE = 22.230 degrees EAST LONGITUDE = 312.070 degrees
TEMPERATURE = 168.5 K
DENSITY (Low, Avg., High) = .5150E-10 .8682E-10 .1779E-09 kg/m**3
DENSITY PLUS PERTURBATION = .1842E-09 kg/m**3
PRESSURE = .3323E-05 N/m**2
EASTWARD WIND = 47.1 m/s NORTHWARD WIND = 33.6 m/s

TIME (relative to initial time) = .0 seconds
HEIGHT = 100.0 km SCALE HEIGHT = 7.64 km
LATITUDE = 22.230 degrees EAST LONGITUDE = 312.070 degrees
TEMPERATURE = 139.8 K
DENSITY (Low, Avg., High) = .2191E-07 .2642E-07 .4082E-07 kg/m**3
DENSITY PLUS PERTURBATION = .4446E-07 kg/m**3
PRESSURE = .7089E-03 N/m**2
EASTWARD WIND = 26.9 m/s NORTHWARD WIND = -2.6 m/s

TIME (relative to initial time) = .0 seconds
HEIGHT = 50.0 km SCALE HEIGHT = 6.87 km
LATITUDE = 22.230 degrees EAST LONGITUDE = 312.070 degrees

```
TEMPERATURE =    129.3 K
DENSITY (Low, Avg., High) =    .2945E-04   .3577E-04   .5575E-04 kg/m**3
DENSITY PLUS PERTURBATION =    .5283E-04 kg/m**3
PRESSURE =    .8884     N/m**2
EASTWARD WIND =     12.1 m/s     NORTHWARD WIND =    -1.2 m/s
------------------------------------------------------------------------
TIME (relative to initial time) =           .0 seconds
HEIGHT =       .0 km              SCALE HEIGHT =  10.48 km
LATITUDE =   22.230 degrees      EAST LONGITUDE =  312.070 degrees
TEMPERATURE =   203.3 K
DENSITY (Low, Avg., High) =    .1193E-01   .1327E-01   .1495E-01 kg/m**3
DENSITY PLUS PERTURBATION =    .1465E-01 kg/m**3
PRESSURE =    517.9     N/m**2
EASTWARD WIND =       .5 m/s     NORTHWARD WIND =       .0 m/s
------------------------------------------------------------------------
```

TABLE B-3

PRINTOUT OF MARSOUT OUTPUT FILE

600.0	22.23	312.07	-15.49	168.71	.05	.51	118.74
550.0	22.23	312.07	-15.44	168.71	.04	.48	55.17
500.0	22.23	312.07	-15.36	168.71	.04	.45	78.28
450.0	22.23	312.07	-15.21	168.71	.01	.43	140.42
400.0	22.23	312.07	-14.86	168.71	-.11	.44	222.63
350.0	22.23	312.07	-14.27	168.71	-.58	.63	236.50
300.0	22.23	312.07	-13.54	168.71	-2.65	1.59	197.86
250.0	22.23	312.07	-12.74	168.71	-11.44	6.07	155.35
200.0	22.23	312.07	-11.76	168.71	-41.07	34.65	130.06
150.0	22.23	312.07	-10.06	168.47	47.09	33.65	72.78
100.0	22.23	312.07	-7.58	139.77	26.86	-2.58	35.79
50.0	22.23	312.07	-4.45	129.35	12.14	-1.19	36.77
.0	22.23	312.07	-1.88	203.26	.49	-.05	11.39

TABLE B-4

PRINTOUT OF DENSAV FILE

600.000	-15.49
550.000	-15.44
500.000	-15.36
450.000	-15.21
400.000	-14.86
350.000	-14.27
300.000	-13.54
250.000	-12.74
200.000	-11.76
150.000	-10.06
100.000	-7.578
50.000	-4.446
.000	-1.877